
Writing a Delphi Property Editor
Version 1.0, 5 March, 1999

1. Introduction
If you regularly write Delphi components you’ll sooner or later has to write specialised property editors
for some properties. Unfortunately this is rather poorly described in the Delphi manuals. In my case I
found myself regularly using workarounds rather than sitting down and learn how to do it. Fortunately I
now have come to the point where I have no choice. Since memory is short, and other people probably
have the same problems as I have had, I decided to write down my experiences. To my help I had the
following documents: the Delphi help-files, [Borland, 1996], [Gopaniouk, 1998], and [Miano, 1997]. I
wrote in Delphi 4, but most of it should be valid for other versions of Delphi too. At a couple of occasions
I ran into problems which I couldn’t handle myself, and I’d like to take the opportunity to thank the
readers of comp.lang.pascal.delphi.components.writing for helping me with these.

Since this is the first time I write a property editor, and the paper is written simultaneously with the
development of the editor, the paper will likely contain errors, omissions, and obscurities. Please report
any such findings along with comments to me at henrikbe@dsv.su.se. Also because of the way the paper
is written it does contain all the errors I did during the coding. This might be irritating to some, but I feel
that it might be useful to see what kind of errors can occur.

2. The Writing
The problem I had was that I had a component (in this paper called TExampleContainer) which needed to
be associated with one or more components of another type (here called TExampleItems). Of course I
could have used a StringList, and then use FindComponent, but this would be too slow for my liking, plus
that it would prevent me from getting a list of the available TExampleItems to chose from. That is, I
wanted to be able to click on the Items property in TExampleContainer and get a window like the one in
Figure 1.

Figure 1

2.1. Preparations
I started with creating a main form and the visible editor form (the one in Figure 1). Then I started to
create the example components. Since I didn’t want to dabble with too many units I put all of them in the
same unit as the editor form.

Now the first problem occurred, TExampleContainer needed a way to store the selected TExampleItems.
The obvious way to do this is in an array, and since this should be allowed according to [Borland, 1996] I
decided to try it out first. This gave me a unit with the following code:

unit PropertyEditorExampleUnit;

interface

uses
  Windows, Messages, SysUtils, Classes, Graphics, Controls, Forms,
Dialogs,
  StdCtrls;

type
  TExampleItemPropertEditorForm = class(TForm)
    AvailableListBox: TListBox;
    Label1: TLabel;
    UsedListBox: TListBox;
    Label2: TLabel;
    AddButton: TButton;
    RemoveButton: TButton;
    OkButton: TButton;
  private
    { Private declarations }
  public
    { Public declarations }
  end;

  TExampleItem = class (TComponent)
  private
    { Private declarations }
    FI:integer;
  public
    { Public declarations }
    property i: integer read FI write FI;
  end;

  TExampleContainer = class (TComponent)
  private
    { Private declarations }
    FEI: array [1..10] of TExampleItem;

    function GetFEI(Index: integer): TExampleItem;
    procedure SetFEI(index: integer; EI: TExampleItem);
  public
    { Public declarations }
    property ExampleItems[Index: integer]: TExampleItem read GetFEI
write setFEI; default;
  end;

procedure register;

var
  ExampleItemPropertEditorForm: TExampleItemPropertEditorForm;

implementation

{$R *.DFM}

procedure Register;
begin
     RegisterComponents('HB', [TExampleContainer, TExampleItem]);
end;

function TExampleContainer.GetFEI(Index: integer): TExampleItem;
begin
     // Naturally we need checks here etc
     GetFEI:=FEI[Index];
end;

procedure TExampleContainer.SetFEI(index: integer; EI: TExampleItem);
begin
     // Naturally we need checks here etc
     FEI[Index]:=EI;
end;

end.

To see if it worked I tried to create a TExampleContainer in the main forms OnShow:

procedure TMainForm.FormShow(Sender: TObject);
var EC:TExampleContainer;
    EI:TExampleItem;
begin
     EC:=TExampleContainer.Create(Application);
     EI:=TExampleItem.Create(Application);
     EC.ExampleItems[1]:=EI;
     EC.ExampleItems[1].i:=100;
     MainForm.caption:=inttostr(EI.i);
end;

This worked perfectly, so I decided to try to install the components. Unfortunately it didn’t work. After a
lot of searching and debugging I posted a question to the newsgroup
comp.lang.pascal.delphi.components.writing, and the answer turned out to be quite strange. Apparently
Delphi is case sensitive on the register procedure. Changing this declaration to “procedure Register”
solved the problem.

After opening the main form again I placed a TExampleContainer and a number of TExampleItems on it
and clicked on the TExampleContainer, and naturally didn’t see the ExampleItems property since I had
forgot to add the published part to the declarations of both TExampleContainer and TExampleItem. Gosh,

what kind of trouble you get in when you are trying to do things manually. Unfortunately this resulted in
the following irritating error message when I tried to install TExampleContainer again:

[Error] PropertyEditorExampleUnit.pas(30): Published property 'ExampleItems' cannot be of type
ARRAY

So, it didn’t work…  What to do now? Either I could move the property out of the published section, or I
could change it into something that could be published. To move the property out of the published section
would mean that it wouldn’t show up in the object inspector, which we wanted. To change the type would
mean having to write a new class for that. In the end the latter solution was preferred, and a new class
called TExampleItemList was created. The relevant code now looked like:

TExampleItemList = class (TObject)
  private
    { Private declarations }
    FEI: array [1..10] of TExampleItem;

    function GetFEI(Index: integer): TExampleItem;
    procedure SetFEI(index: integer; EI: TExampleItem);
  public
    { Public declarations }
    property ExampleItems[Index: integer]: TExampleItem read GetFEI
write setFEI; default;
  end;

  TExampleContainer = class (TComponent)
  private
    { Private declarations }
    FEIL: TExampleItemList;
  public
    { Public declarations }
  published
    property Items: TExampleItemList read FEIL write FEIL;
  end;

function TExampleItemList.GetFEI(Index: integer): TExampleItem;
begin
     // Naturally we need checks here etc
     GetFEI:=FEI[Index];
end;

procedure TExampleItemList.SetFEI(index: integer; EI: TExampleItem);
begin
     // Naturally we need checks here etc
     FEI[Index]:=EI;
end;

Now, after recompiling the package the whole thing worked. When placing a TExampleContainer and a
couple of TExampleItems on the main form their respective properties appeared in the object inspector as
in Figure 2 below. Unfortunately since the Items property of TExampleContainer is a descendent of

TObject, and we haven’t said anything else, the object inspector thinks that it should provide an
expandable list of the properties of the TExampleItemList, which of course isn’t what we want. Also we
haven’t actually created the instance of TExampleItemList that FEIL should be set to. To do this we add a
constructor and destructor to TExampleContainer.

Figure 2

// Under the public part of TEC
constructor create(AOwner: TComponent); override;
destructor destroy; override;

// The implementation
constructor TExampleContainer.create(AOwner: TComponent);
begin
     inherited create(AOwner);
     FEIL:=TExampleItemList.Create;
end;

destructor TExampleContainer.destroy;
begin
     // I'm not sure if this is really necessary,
     // but lets do it anywhay
     FEIL.destroy;

     inherited destroy;
end;

After the standard routine of recompilation and placement of the component on the main form I tried
clicking on the items property of the TExampleContainer. This gave me an “Access Violation”. Since I
wasn’t sure if this depended on that TExampleItemList don’t contain any published properties or that no
property editor was defined for TExampleItemList I added a simple numerical property to
TExampleItemList. This didn’t solve the problem so now I thought it time to start building the actual
property editor.

2.2. Writing the basic property editor
The first problem was to choose a suitable parent for the editor. According to the help files the following
ones where available apart from the base class TPropertyEditor:

Type Properties edited

TOrdinalProperty All ordinal-property editors (those for integer, character, and enumerated
properties) descend from TOrdinalProperty.

TIntegerProperty All integer types, including predefined and user-defined subranges.

TCharProperty Char-type and subranges of Char, such as ‘A’..’Z’.

TEnumProperty Any enumerated type.

TFloatProperty All floating-point numbers.

TStringProperty Strings.

TSetElementProperty Individual elements in sets, shown as Boolean values

TSetProperty All sets. Sets are not directly editable, but can expand into a list of set-element
properties.

TClassProperty Classes. Displays the name of the class and allows expansion of the class’s
properties.

TMethodProperty Method pointers, most notably events.

TComponentProperty Components in the same form. The user cannot edit the component’s properties,
but can point to a specific component of a compatible type.

TColorProperty Component colors. Shows color constants if applicable, otherwise displays
hexadecimal value. Drop-down list contains the color constants. Double-click
opens the color-selection dialog box.

TFontNameProperty Font names. The drop-down list displays all currently installed fonts.

TFontProperty Fonts. Allows expansion of individual font properties as well as access to the font
dialog box.

None of these actually seemed to suit my needs, so I choose TPropertyEditor as the base class. I added
“dsgnintf” to the uses clause and set out on my mission to construct my first property editor. The result
can be seen below.

TExampleItemListPropertyEditor = class(TPropertyEditor)
  public
        function GetValue:string; override;
        function GetAttributes:TPropertyAttributes; override;
        procedure Edit; override;
  end;

function TExampleItemListPropertyEditor.GetValue:string;
var n:integer;
begin
     result:='';
     for n:=1 to 10 do begin
         if (GetComponent(0) as
TExampleContainer).items.ExampleItems[n]<> nil then
            result:=result+(GetComponent(0) as TComponent).name+' ';
     end;
end;

function
TExampleItemListPropertyEditor.GetAttributes:TPropertyAttributes;

begin
     result:=[paDialog];
end;

procedure TExampleItemListPropertyEditor.Edit;
var form:TExampleItemPropertEditorForm;
begin
     form := TExampleItemPropertEditorForm.create(Application);
     try
        form.showModal;
     finally
         form.Destroy;
     end;
end;
I also changed the Register-procedure to look like:

procedure Register;
begin
     RegisterComponents('HB', [TExampleContainer, TExampleItem]);
     RegisterPropertyEditor(
                            TypeInfo(TExampleItemList),
                            nil, '',TExampleItemListPropertyEditor);
end;

After recompilation of the library etc I tried it out again, and now, a miracle occurred, it actually worked.
More and less in the first try too. Now all that was needed was to add the actual functionality to the
property editor (I thought).

2.3. Adding functionality
There were two things necessary to do to add the desired functionality to the property editor. I needed to
establish contact with the component which the user was editing, and I needed to get a list of all
TExampleItems on the form. The first of these wasn’t much of a problem. The GetComponent in
TPropertyEditor was capable of doing that. The second needed some more thought though. The only way
I could think of was to use the components' owner property and there iterate over all items in the
components array. This resulted in the following code:

procedure TExampleItemListPropertyEditor.Edit;
var form:TExampleItemPropertEditorForm;
    n:integer;
begin
     form := TExampleItemPropertEditorForm.create(Application);
     try
        // Write the available TExampleItems into the
        // left listbox
        with (GetComponent(0) as TComponent).owner do begin
             for n:=0 to componentCount-1 do begin
                 if Components[n] is TExampleItem then begin
                   form.AvailableListBox.items.add(Components[n].name);
                 end;
             end;

        end;

        form.showModal;

        // Add the TExampleItems in the right listbox
        // to the TExampleContainer
        with form.UsedListBox do begin
             for n:=1 to Items.Count do begin
                 (GetComponent(0) as
TExampleContainer).Items.ExampleItems[n-1]:=(findcomponent(items[n-1])
as TExampleItem);
             end;
        end;
     finally
         form.Destroy;
     end;
end;

Well, now things looked good, the code worked, and I could choose the TExampleItems in the property
editor. But, when I tested the application I noticed that the TExampleItem-references weren’t saved. After
much debugging, I concluded that they weren’t streamed to the dfm-file and set about trying to get this to
work. The help files provided little or no help, so I tried a new Usenet call, which unfortunately didn’t
receive any answers. So finally I did what I should have done a long time ago, I went out and bought a
book on the subject, [Teixeira and Pacheco, 1998].  Following their example I added the following
methods to TExampleItemList.

procedure ReadFEIData(Reader: TReader);
procedure WriteFEIData(Writer: TWriter);
procedure DefineProperties(Filer: TFiler); // Protected

Then I ran into another dead end, I didn’t know how to read and write the FEI from and to a stream. Since
the components already exists, I only want to handle references to them, but how do you do this so that the
references work the next time you open the project? Two days later, after another Usenet question with
better luck, it all seemed so obvious. The missing piece of information that I got was that Delphi stores the
names of the components, and resolves the addresses after all components has been loaded.

procedure TExampleItemList.ReadFEIData(Reader: TReader);
var n:integer;
begin
     for n:=1 to 10 do begin
         FEIStr:=Reader.ReadString;
     end;
end;

procedure TExampleItemList.WriteFEIData(Writer: TWriter);
var n:integer;
    str:string;
begin
     for n:=1 to 10 do begin
         if FEI[n]<>nil then str:=str+FEI[n].name;

         str:=str+' ';
     end;
     Writer.WriteString(str);
end;

procedure TExampleItemList.DefineProperties(Filer: TFiler);
begin
     inherited DefineProperties(Filer);

     // I could of course had saved each of the FEI:s
     // in turn, but that would be tedious. In that case
     // I would probably have saved only the parts where
     // there were data
     Filer.DefineProperty('FEI',ReadFEIData,WriteFEIData,true);
end;

Now we save a list of names separated with spaces. The only thing that needs to be done now is to parse
this string when all components are available on the form. To do this I override the method “Loaded” in
TExampleContainer with the following code. The code is a bit ugly since I break the encapsulation
principle by accessing a private member of TExampleItemList, but it saved me a method call, and Delphi
allows it since they are in the same unit so lets do it the quick and dirty way.

procedure TExampleContainer.Loaded;
var n:integer;
    str,old:string;
begin
     old:=FEIL.FEIStr;
     for n:=1 to 10 do begin
         // Parse the string
         str:=copy(old, 1, pos('#', old)-1);
         old:=copy(old, Pos('#', old)+1, maxint);
         // Find the TExampelItem with the same name
         FEIL.FEI[n]:=(owner.FindComponent(str) as TExampleItem);
     end;
     inherited loaded;
end;

This concludes this little experience report. It took me more time than I had expected, but in the end it
worked out anyway. The methods used to solve problems here goes along the line “if it’s stupid but
works, it ain’t stupid”, and the code would need much improvement before it is anywhere near finished.
But that has to be left to others since I now have to start using what I learned during the time I wrote this
program. If someone finds errors, omissions etc in this paper I would appreciate if they brought it tom my
attention. The same goes for better ways to solve the problems I experienced, such as finding components
etc.

05 March 1999

Henrik Bergström

Appendix – Full Source Code
unit PropertyEditorExampleUnit;

interface

uses
// Contains the property editor classes
dsgnintf,
Windows, Messages, SysUtils, Classes, Graphics, Controls, Forms,
Dialogs,StdCtrls;

type
  TExampleItem = class (TComponent)
  private
    { Private declarations }
    FI:integer;
  public
    { Public declarations }
  published
    property i: integer read FI write FI;
  end;

  TExampleItemList = class (TPersistent)
  private
    // Private declarations
    FEIStr: string;
    FEI: array [1..10] of TExampleItem;
    FI: integer;

   function GetFEI(Index: integer): TExampleItem;
    procedure SetFEI(index: integer; EI: TExampleItem);

    procedure ReadFEIData(Reader: TReader);
    procedure WriteFEIData(Writer: TWriter);

  protected
    procedure DefineProperties(Filer: TFiler); override;



Henrik Bergström
DSV SU/KTH
1999-03-05

11

  public
    // Public declarations
    property ExampleItems[Index: integer]: TExampleItem
             read GetFEI write setFEI; default;

  published
    // Test property added to see where the
    // Access Violation came from
    property i: integer read FI write FI;
  end;

  TExampleContainer = class (TComponent)
  private
    { Private declarations }
    FEIL: TExampleItemList;
    FI:integer;
  public
    { Public declarations }
    constructor create(AOwner: TComponent); override;
    destructor destroy; override;
  protected
     procedure Loaded; override;
  published
    property Items: TExampleItemList read FEIL write FEIL;
    property i: integer read FI write FI;
  end;

  TExampleItemListPropertyEditor = class(TPropertyEditor)
  public
        function GetValue:string; override;
        function GetAttributes:TPropertyAttributes; override;
        procedure Edit; override;
  end;

  TExampleItemPropertEditorForm = class(TForm)
    AvailableListBox: TListBox;
    Label1: TLabel;
    UsedListBox: TListBox;
    Label2: TLabel;
    AddButton: TButton;
    RemoveButton: TButton;
    OkButton: TButton;
    procedure AddButtonClick(Sender: TObject);
    procedure RemoveButtonClick(Sender: TObject);
    procedure UsedListBoxDblClick(Sender: TObject);
  private
    { Private declarations }

  public
    { Public declarations }
  end;

procedure Register;

var
  ExampleItemPropertEditorForm: TExampleItemPropertEditorForm;

implementation

{$R *.DFM}

procedure Register;
begin
     RegisterComponents('HB', [TExampleContainer, TExampleItem]);
     RegisterPropertyEditor(
                            TypeInfo(TExampleItemList),
                            nil, '',TExampleItemListPropertyEditor);
end;

constructor TExampleContainer.create(AOwner: TComponent);
begin
     inherited create(AOwner);
     FEIL:=TExampleItemList.Create;
end;

destructor TExampleContainer.destroy;
begin
     FEIL.destroy;
     inherited destroy;
end;

procedure TExampleContainer.Loaded;
var n:integer;
    str,old:string;
begin
     old:=FEIL.FEIStr;
     for n:=1 to 10 do begin
         // Parse the string
         str:=copy(old, 1, pos('#', old)-1);
         old:=copy(old, Pos('#', old)+1, maxint);
         // Find the TExampelItem with the same name
         FEIL.FEI[n]:=(owner.FindComponent(str) as TExampleItem);
     end;
     inherited loaded;
end;

function TExampleItemList.GetFEI(Index: integer): TExampleItem;
begin
     // Naturally we need checks here etc
     GetFEI:=FEI[Index];

end;

procedure TExampleItemList.SetFEI(index: integer; EI: TExampleItem);
begin
     // Naturally we need checks here etc
     FEI[Index]:=EI;
end;

procedure TExampleItemList.ReadFEIData(Reader: TReader);
begin
     FEIStr:=Reader.ReadString;
end;

procedure TExampleItemList.WriteFEIData(Writer: TWriter);
var n:integer;
    str:string;
begin
     for n:=1 to 10 do begin
         if FEI[n]<>nil then str:=str+FEI[n].name;
         str:=str+'#';
     end;
     Writer.WriteString(str);
end;

procedure TExampleItemList.DefineProperties(Filer: TFiler);
begin
     inherited DefineProperties(Filer);

     // I could of course had saved each of the FEI:s
     // in turn, but that would be tedious. In that case
     // I would probably have saved only the parts where
     // there were data
     Filer.DefineProperty('FEI',ReadFEIData,WriteFEIData,true);
end;

function TExampleItemListPropertyEditor.GetValue:string;
var n:integer;
begin
     result:='';
     for n:=1 to 10 do begin
         if (GetComponent(0)
              as TExampleContainer).items.ExampleItems[n]<> nil then
            result:=result+
(GetComponent(0) as TExampleContainer).items.ExampleItems[n].name+' ';
     end;
end;

function TExampleItemListPropertyEditor.GetAttributes:
 TPropertyAttributes;

begin
     result:=[paDialog];

end;

procedure TExampleItemListPropertyEditor.Edit;
var form:TExampleItemPropertEditorForm;
    n:integer;
begin
     form := TExampleItemPropertEditorForm.create(Application);
     try
        // Write the available TExampleItems into the
        // left listbox
        with (GetComponent(0) as TComponent).owner do begin
             for n:=0 to componentCount-1 do begin
                 if Components[n] is TExampleItem then begin
                   form.AvailableListBox.items.add(Components[n].name);
                 end;
             end;
        end;

        form.showModal;

        // Add the TExampleItems in the right listbox
        // to the TExampleContainer
        with form.UsedListBox do begin
             for n:=1 to Items.Count do begin
                 messagedlg(((GetComponent(0) as
TComponent).owner.findcomponent(items[n-1]) as TExampleItem).name,
mtInformation, [mbOk],0);
                 (GetComponent(0) as
TExampleContainer).Items.ExampleItems[n]:=((GetComponent(0) as
TComponent).owner.findcomponent(items[n-1]) as TExampleItem);
             end;
        end;
        // Might be necessary according to the FAQ
        Designer.modified;
     finally
         form.Destroy;
     end;
end;

procedure TExampleItemPropertEditorForm.AddButtonClick(Sender:
TObject);
begin

UsedListBox.Items.add(AvailableListBox.items[AvailableListBox.itemindex
]);
     AvailableListBox.items.Delete(AvailableListbox.itemindex);
end;

procedure TExampleItemPropertEditorForm.RemoveButtonClick(Sender:
TObject);

begin

AvailableListBox.Items.add(UsedListBox.items[UsedListBox.itemindex]);
     UsedListBox.items.Delete(UsedListbox.itemindex);
end;

procedure TExampleItemPropertEditorForm.UsedListBoxDblClick(
  Sender: TObject);
begin
     RemoveButtonClick(nil);
end;

end.


